

PROGRAMMING CHALLENGES USING ARRAYS

1. Store your own tune data in the SOUNDARY program. READ it into an array and play different sequences of notes from the main loop.
2. Give the person using your program the option to edit his or her tune, one note at a time. Ask the person which note he or she wants to change and use that number as the index to your array. Then ask what the new note value will be. Change the array element accordingly and return to the menu.
3. Write a program which INPUTs a series of numbers, and then lists those numbers which are larger than the average of all the numbers which were typed in. Use an array to store the values which the person types in.
4. Use two different arrays, HUE(100) and LUM(100), to create a light show. Your program should contain a statement like the following one:

```
SETCOLOR 0,HUE(COUNT),LUM(COUNT)
```

ARRAYS
CAMPER COPY

```
100 REM *      ARRAY OF NUMBERS
110 REM *
120 MAXNUMS = 5 :REM MAXIMUM NUMBERS WHICH CAN BE INPUT
130 DIM NUMS(MAXNUMS)
140 REM *
150 REM *      FILL ARRAY
160 REM *
170 FOR COUNT = 1 TO MAXNUMS
180 PRINT "TYPE IN A NUMBER";
190 INPUT VALUE
200 NUMS(COUNT) = VALUE :REM STORE INPUT VALUE IN ARRAY
210 NEXT COUNT
```

```
220 REM *
230 REM *      PRINT NUMBERS IN ORDER
240 REM *
250 PRINT
260 PRINT "YOU TYPED IN THE NUMBERS IN THE"
270 PRINT "FOLLOWING ORDER:"
280 PRINT
290 FOR COUNT = 1 TO MAXNUMS
300 PRINT NUMS(COUNT);" "; :REM PRINT VALUE + SPACES
310 NEXT COUNT
320 PRINT
```

```
100 REM *      SOUND WITH AN ARRAY
110 REM *
120 DIM TUNE(100)
130 XNOTE = 0
140 INPUT PITCH
150 IF PITCH = -1 THEN NUMNOTES = XNOTE:GOTO 200
160 XNOTE = XNOTE + 1
170 TUNE(XNOTE) = PITCH
180 GOTO 140
```

0000 0000 0000 0000
0000 0000 0000 0000

0000 0000 0000 0000
0000 0000 0000 0000
0000 0000 0000 0000
0000 0000 0000 0000
0000 0000 0000 0000

0000 0000
0000 0000
0000 0000

0000
0000
0000

ARRAYS
CAMPER COPY CONTINUED

```
100 REM *      SOUND ARRAY
110 REM *
120 REM *
130 REM *      INITIALIZE VARIABLES AND ARRAY
140 REM *
150 DIM TUNE(100)
160 XNOTE=0
165 REM * ASSIGN LABELS TO LINE NUMBERS
170 MENU=300
180 VALUES=500
190 PLAY=700
200 NUMBERS=900
210 REM *
220 REM *      MAIN LOOP
230 REM *
240 GOSUB MENU
250 INPUT RESPONSE
260 IF RESPONSE=1 THEN GOSUB VALUES
270 IF RESPONSE=2 THEN GOSUB PLAY
280 IF RESPONSE=3 THEN GOSUB NUMBERS
290 GOTO 240:REM REPEAT MAIN LOOP
300 REM *
310 REM *      MENU
320 REM *
330 PRINT
340 PRINT "WOULD YOU LIKE TO:"
350 PRINT "      1.  TYPE IN A TUNE."
360 PRINT "      2.  PLAY YOUR TUNE."
370 PRINT "      3.  LIST THE NOTES."
380 PRINT
390 PRINT "TYPE IN A NUMBER";
400 PRINT :REM INPUT IN MAIN LOOP
410 RETURN
500 REM *
510 REM *      INPUT VALUES FOR NOTES
520 REM *
530 PRINT " TYPE IN NUMBERS BETWEEN 0"
540 PRINT " AND 255 TO BE THE NOTES"
550 PRINT " OF A TUNE.  TYPE ONE NOTE"
560 PRINT " PER ?.  WHEN YOU ARE FINISHED,"
570 PRINT " TYPE A -1 FOR THE LAST NOTE."
580 INPUT PITCH
590 IF PITCH>255 OR PITCH<-1 THEN 580
600 REM * MINUS ONE IS A FLAG FOR THE END OF THE DATA
610 IF PITCH=-1 THEN NUMNOTES=XNOTE:RETURN
620 XNOTE=XNOTE+1:REM NOTES COUNTER
630 TUNE(XNOTE)=PITCH
640 GOTO 580
```

ARRAYS
CAMPER COPY CONTINUED

```
700 REM *
710 REM *      PLAY TUNE
720 REM *
730 FOR XNOTE=1 TO NUMNOTES
740 SOUND 0,TUNE(XNOTE),10,10
750 FOR DELAY=1 TO 10:NEXT DELAY
760 NEXT XNOTE
770 SOUND 0,0,0,0
780 RETURN
900 REM *
910 REM *      LIST NOTES
920 REM *
930 FOR XNOTE=1 TO NUMNOTES
940 PRINT "TUNE(";XNOTE;")";" ";TUNE(XNOTE)
950 NEXT XNOTE
960 RETURN
```

ARRAYS
CAMPER COPY CONTINUED

TUNE

```
100 REM *          TUNE ARRAY
110 REM *
120 DIM PITCH(50),DISTORT(50),LOUD(50),TIME(50)
130 INIT=500:REM INITIALIZATION LINE#
140 PLAY=300:REM PLAY TUNE ROUTINE
150 MAXNOTES=11
200 REM *
210 REM ***** MAIN LOOP *****
220 REM *
230 GOSUB INIT
240 START=1:FINISH=5:GOSUB PLAY
250 START=6:FINISH=11:GOSUB PLAY
260 START=1:FINISH=4:GOSUB PLAY
270 END
300 REM
310 REM ***** PLAY *****
320 REM *
330 REM *  PLAYS A SEQUENCE OF NOTES USING DATA ARRAYS.
340 REM *  INDICES DETERMINED BY VALUES OF START AND
350 REM *  FINISH IN MAIN LOOP
360 REM *
370 FOR XNOTE=START TO FINISH
380 SOUND 0,PITCH(XNOTE),DISTORT(XNOTE),LOUD(XNOTE)
390 FOR DELAY=1 TO TIME(XNOTE):NEXT DELAY
400 NEXT XNOTE
410 RETURN
420 REM *
500 REM ***** INIT ARRAY *****
510 REM *
520 FOR FILL=1 TO MAXNOTES
530 READ PITCH,DISTORT,LOUD,TIME
540 PITCH(FILL)=PITCH:DISTORT(FILL)=DISTORT:LOUD(FILL)=LOUD:TIME(FILL)=TIME
550 NEXT FILL
560 RETURN
570 DATA 121,10,10,40,91,10,10,37,0,0,0,3,91,10,10,40,108,10,10,28
580 DATA 0,0,0,2,108,10,10,10,91,10,10,30,108,10,10,10,121,10,10,80,0,0,0,0
```